Dynamics Forward/Backwards

```
> restart:
  with(Student[LinearAlgebra]):
  with(Physics):
  Setup(mathematicalnotation=true):
```

Lagrangian for XZ plane (forwards/backwards motion)

```
> Lagr := (1/2)*M*((L*diff(theta(t),t))^2+2*L*diff(theta(t),t)*diff
  (x(t),t)+diff(x(t),t)^2)+(1/2)*Iyy_g*diff(theta(t),t)^2 + M*g*L*
  theta(t)^2/2;
```

$$Lagr := \frac{M\left(L^2\,\dot{\theta}(t)^2+2\,L\,\dot{\theta}(t)\,\dot{x}(t)+\dot{x}(t)^2\right)}{2}+\frac{Iyy\_g\,\dot{\theta}(t)^2}{2}+\frac{M\,g\,L\,\theta(t)^2}{2} \tag{1.1}$$

```
> eq1 := diff(diff(Lagr,diff(theta(t),t)),t) - diff(Lagr,theta(t))
  = 0;
  eq2 := diff(diff(Lagr,diff(x(t),t)),t) - diff(Lagr,x(t)) = F(t);
  F(t) := (T(t)-I_w*diff(x(t),t,t)/R_w)/R_w-m_w*diff(x(t),t,t);
  T(t) := (K_t/R)*(V(t)-K_phi*(diff(x(t),t)/R_w));
  sys := {eq1, eq2}:
```

$$eq1 := \frac{M\left(2\,L^2\,\ddot{\theta}(t)+2\,L\,\ddot{x}(t)\right)}{2}+Iyy\_g\,\ddot{\theta}(t)-M\,g\,L\,\theta(t)=0$$

$$eq2 := \frac{M\left(2\,L\,\ddot{\theta}(t)+2\,\ddot{x}(t)\right)}{2}=F(t)$$

$$F(t) := \frac{T(t)-\frac{I\_w\,\ddot{x}(t)}{R\_w}}{R\_w}-m\_w\,\ddot{x}(t)$$

$$T(t) := \frac{K\_t\left(V(t)-\frac{K\_phi\,\dot{x}(t)}{R\_w}\right)}{R} \tag{1.2}$$

```
> sol := simplify(solve(sys,[diff(x(t),t,t),diff(theta(t),t,t)]))
```

$$sol := \Bigg[\Bigg[\ddot{x}(t) = \Big(-K\_phi\,K\_t\left(L^2\,M+Iyy\_g\right)\dot{x}(t)+\Big(K\_t\left(L^2\,M+Iyy\_g\right)V(t) - L^2\,\theta(t)\,M^2\,R\,g\,R\_w\Big)\,R\_w\Big)\Big/\Big(\Big(\Big(\left(L^2\,m\_w+Iyy\_g\right)M+Iyy\_g\,m\_w\Big)R\_w^2 + I\_w\left(L^2\,M+Iyy\_g\right)\Big)R\Big),\ \ddot{\theta}(t) = \frac{\left(K\_phi\,K\_t\,\dot{x}(t)+g\,R\left(\left(M+m\_w\right)R\_w^2+I\_w\right)\theta(t)-K\_t\,V(t)\,R\_w\right)M\,L}{\left(\left(\left(L^2\,m\_w+Iyy\_g\right)M+Iyy\_g\,m\_w\right)R\_w^2+I\_w\left(L^2\,M+Iyy\_g\right)\right)R}\Bigg]\Bigg] \tag{1.3}$$

A and B for xdotdot

```
> A21:=diff(rhs(sol[1][1]), x(t));
  A22:=diff(rhs(sol[1][1]), diff(x(t),t));
  A23:=diff(rhs(sol[1][1]), theta(t));
```

```
A24:=diff(rhs(sol[1][1]), diff(theta(t),t));
B2:=diff(rhs(sol[1][1]), V(t));
```

$$A21 := 0$$

$$A22 := -\frac{K\_phi\,K\_t\,(L^2 M + Iyy\_g)}{(((L^2 m\_w + Iyy\_g)\,M + Iyy\_g\,m\_w)\,R\_w^2 + I\_w\,(L^2 M + Iyy\_g))\,R}$$

$$A23 := -\frac{L^2 M^2 g\,R\_w^2}{((L^2 m\_w + Iyy\_g)\,M + Iyy\_g\,m\_w)\,R\_w^2 + I\_w\,(L^2 M + Iyy\_g)}$$

$$A24 := 0$$

$$B2 := \frac{K\_t\,(L^2 M + Iyy\_g)\,R\_w}{(((L^2 m\_w + Iyy\_g)\,M + Iyy\_g\,m\_w)\,R\_w^2 + I\_w\,(L^2 M + Iyy\_g))\,R} \quad \textbf{(1.4)}$$

A and B for thetadotdot

```
> A41:=diff(rhs(sol[1][2]), x(t));
A42:=diff(rhs(sol[1][2]), diff(x(t),t));
A43:=diff(rhs(sol[1][2]), theta(t));
A44:=diff(rhs(sol[1][2]), diff(theta(t),t));
B4:=diff(rhs(sol[1][2]), V(t));
```

$$A41 := 0$$

$$A42 := \frac{K\_phi\,K\_t\,M\,L}{(((L^2 m\_w + Iyy\_g)\,M + Iyy\_g\,m\_w)\,R\_w^2 + I\_w\,(L^2 M + Iyy\_g))\,R}$$

$$A43 := \frac{g\,((M + m\_w)\,R\_w^2 + I\_w)\,M\,L}{((L^2 m\_w + Iyy\_g)\,M + Iyy\_g\,m\_w)\,R\_w^2 + I\_w\,(L^2 M + Iyy\_g)}$$

$$A44 := 0$$

$$B4 := -\frac{K\_t\,R\_w\,M\,L}{(((L^2 m\_w + Iyy\_g)\,M + Iyy\_g\,m\_w)\,R\_w^2 + I\_w\,(L^2 M + Iyy\_g))\,R} \quad \textbf{(1.5)}$$

Dynamics Sideways

```
> restart:
with(Student[LinearAlgebra]):
with(Physics):
Setup(mathematicalnotation=true):
```

Lagrangian for YZ plane (sideways motion)

```
> Lagr := (1/2)*(Ixx_g + M*(L+R_w)^2)*diff(phi(t),t)^2+M*g*(L+R_w)*
phi(t)^2/2;
```

$$Lagr := \frac{(Ixx\_g + M\,(L + R\_w)^2)\,\dot{\phi}(t)^2}{2} + \frac{M g\,(L + R\_w)\,\phi(t)^2}{2} \quad \textbf{(2.1)}$$

Torque exerted on flywheel and thus -Torque exerted on the robot

```
> eq1 := diff(diff(Lagr,diff(phi(t),t)),t) - diff(Lagr,phi(t)) = T
(t);
eq2 := T(t) = -I_w*diff(alpha(t),t,t);
T(t) := (K_t/R)*(V(t)-K_phi*diff(alpha(t),t));
sys := {eq1, eq2}:
```

$$eq1 := (Ixx\_g + M\,(L + R\_w)^2)\,\ddot{\phi}(t) - M g\,(L + R\_w)\,\phi(t) = T(t)$$

$$eq2 := T(t) = -I\_w\, \ddot{\alpha}(t)$$

$$T(t) := \frac{K\_t\, (V(t) - K\_phi\, \dot{\alpha}(t))}{R} \tag{2.2}$$

```
> sol := simplify(solve(sys,[diff(phi(t),t,t), diff(alpha(t),t,t)])
  )
```

$$sol := \left[\left[\ddot{\phi}(t) = \frac{-K\_phi\, K\_t\, \dot{\alpha}(t) + M\, R\, g\, (L + R\_w)\, \phi(t) + K\_t\, V(t)}{\left(Ixx\_g + M\, (L + R\_w)^2\right) R}, \ddot{\alpha}(t) = \frac{K\_t\, (K\_phi\, \dot{\alpha}(t) - V(t))}{I\_w\, R}\right]\right] \tag{2.3}$$

```
> A21:=diff(rhs(sol[1][1]), phi(t));
  A22:=diff(rhs(sol[1][1]), diff(phi(t),t));
  A23:=diff(rhs(sol[1][1]), alpha(t));
  A24:=diff(rhs(sol[1][1]), diff(alpha(t),t));
  B2:=diff(rhs(sol[1][1]), V(t));
```

$$A21 := \frac{M\, g\, (L + R\_w)}{Ixx\_g + M\, (L + R\_w)^2}$$

$$A22 := 0$$

$$A23 := 0$$

$$A24 := -\frac{K\_phi\, K\_t}{\left(Ixx\_g + M\, (L + R\_w)^2\right) R}$$

$$B2 := \frac{K\_t}{\left(Ixx\_g + M\, (L + R\_w)^2\right) R} \tag{2.4}$$

```
> A41:=diff(rhs(sol[1][2]), phi(t));
  A42:=diff(rhs(sol[1][2]), diff(phi(t),t));
  A43:=diff(rhs(sol[1][2]), alpha(t));
  A44:=diff(rhs(sol[1][2]), diff(alpha(t),t));
  B4:=diff(rhs(sol[1][2]), V(t));
```

$$A41 := 0$$

$$A42 := 0$$

$$A43 := 0$$

$$A44 := \frac{K\_t\, K\_phi}{I\_w\, R}$$

$$B4 := -\frac{K\_t}{I\_w\, R} \tag{2.5}$$

```
>
```